Math 262a, Fall 1999, Glenn Tesler
Homework 7
11/19/99

```
> with(Ore_algebra):
  cleanpol := (f,Sn) ->
  sort(collect(expand(f),Sn,factor),[Sn,x,n]):
```

Problem 1.

Set up an algebra.

```
> An := shift_algebra([Sn,n]);
```

$$An := Ore\_algebra$$

Express the recurrences in operator notation.

```
> recF := Sn^2 - (n+3)*Sn + 2*n;
  recG := Sn^2 - (2*n+1)*Sn + n^2;
```

$$recF := Sn^2 - (n+3)\,Sn + 2\,n$$

$$recG := Sn^2 - (2\,n+1)\,Sn + n^2$$

Find the gcd and lcm of the operators.

```
> G := skew_gcdex(recF,recG,Sn,An);
```

$$G := [2\,n + Sn\,n - 2\,Sn - n^2, 1, \text{-}1, n^2 - n - Sn\,n + 2\,Sn, -2\,n + 2 + Sn\,n - 2\,Sn]$$

Decode this into gcd and lcm:

```
> gcdFG := cleanpol(G[1],Sn);
  lcmFG := cleanpol(skew_product(G[4],recF,An),Sn);
```

$$gcdFG := (n-2)\,Sn - (n-2)\,n$$

$lcmFG :=$

$$(-n+2)\,Sn^3 + (2\,n^2 + n - 8)\,Sn^2 + 2\,(n-1)\,n^2 + (-n^3 - 4\,n^2 + 5\,n + 4)\,Sn$$

(a) The lcm is the operator notation for the minimal annihilator satisfied by both functions. In function notation it is

```
> applyopr(lcmFG,a(n),An) = 0;
```

$$2\,(n-1)\,n^2\,\mathrm{a}(n) + (-n^3 - 4\,n^2 + 5\,n + 4)\,\mathrm{a}(n+1) + (2\,n^2 + n - 8)\,\mathrm{a}(n+2) + (-n+2)\,\mathrm{a}(n+3) = 0$$

(b) If indeed f(n)=g(n), then the gcd of the two operators is an operator annihilating f(n), and rational multiples of it are too, except their poles/zeros may give exceptional values of n. Here, the gcd is (n-2)(Sn-n), so the function satisfies (Sn-n)f(n)=0 except possibly at n=2. Thus

```
> f(n)=C*(n-1)!;
```

$$f(n) = C\,(n-1)!$$

for n>=1, for some constant C, with a possible exception at n=2. However, setting n=1 in the original recurrence for f(n) gives

```
> subs(n=1,applyopr(recF,f(n),An))=0;
```

$$2\,f(1) - 4\,f(2) + f(3) = 0$$

and the values f(1)=C*0!=C, f(3)=C*2!=2C, force f(2)=C=C*(2-1)!, so it's still true at n=2.

(c) The Euclidean algorithm produced lcm = u*recF = v*recG. The way we did it in class, u and v could have denominators (functions of n), but this would be bad if we divided by 0. Chyzak's implementation clears denominators, resulting in extra factors in the lcm. A complete procedure would be to find the lcm of some order D, s.t. u,v have no denominators. Here, D=3. Verify the first D initial conditions, f(1)=g(1), f(2)=g(2), f(3)=g(3). Then, all further values of f(n), g(n) are equal by iterating the recursion, EXCEPT we cannot deduce f(N+D)=g(N+D) when the leading term of the lcm has a nonnegative integer root N, so we must check this separately. Here this happens for N=2, so we also must check f(5)=g(5).

(d) Using the common right factor Sn-n gives

```
> divF:=skew_pdiv(recF,Sn-n,Sn,An);
  divG:=skew_pdiv(recG,Sn-n,Sn,An);
```

$$divF := [1, Sn-2, 0]$$

$$divG := [1, Sn-n, 0]$$

meaning 1*recF - (Sn-2)*(Sn-n) = 0 (the reason for the multiple of recF is again to clear denominators; instead of 1 it could be a function of n, but is still n-free). So this says recF = (Sn-2)(Sn-n) and recG=(Sn-n)(Sn-n). Check it:

```
> skew_product(Sn-2,Sn-n,A); simplify("-recF);
```

$$2\,n + (-n-3)\,Sn + Sn^2$$

$$0$$

```
> skew_product(Sn-n,Sn-n,A); simplify("-recG);
```

$$n^2 + (-2\,n-1)\,Sn + Sn^2$$

$$0$$

```
>
```

Problem 2. The old way is with Wronskians:

```
> wronsk := proc(f,D,A)
      local n,m,i,j;
      n := nops(f);
```

```
      m := linalg[matrix](n,n);
      for i from 1 to n do
      for j from 0 to n-1 do
          m[i,j+1] := applyopr(skew_power(D,j,A),f[i],A);
      od od;
      RETURN(evalm(m))
  end:
  wronskeq := proc(f,D,A)
      local W, yy,n,eqn;
      W := wronsk(f,D,A);
      n := nops(f);
      yy := ['W[1,n+1-j]'$j=1..n];
      eqn := linalg[det](W);

      # clear the leading coefficient & simplify
      eqn := eqn / coeff(eqn,yy[1]);
      eqn := simplify(eqn); eqn := numer(eqn);

      # collect it and clean it up
      eqn := sort(collect(eqn,{op(yy)},factor),yy) = 0;
  end:
> Ax := diff_algebra([Dx,x]): An := shift_algebra([Sn,n]):
> W := wronsk([y(x),sin(x),x],Dx,Ax);
```

$$W := \begin{bmatrix} y(x) & \frac{\partial}{\partial x} y(x) & \frac{\partial^2}{\partial x^2} y(x) \\ \sin(x) & \cos(x) & -\sin(x) \\ x & 1 & 0 \end{bmatrix}$$

Minimal order diffeq possible

```
> wronskeq([y(x),sin(x),x],Dx,Ax);
```

$$(-\sin(x) + x\cos(x))\left(\frac{\partial^2}{\partial x^2} y(x)\right) + x\sin(x)\left(\frac{\partial}{\partial x} y(x)\right) - \sin(x)\, y(x) = 0$$

Or, we know the minimal diffeq with rational coeffs having sin(x) as a solution is y''+y=0, and cos(x) is another solution of it. So the minimal equation with only rational functions of x as coefficients will be

```
> wronskeq([y(x),sin(x),cos(x),x],Dx,Ax);
```

$$x\left(\frac{\partial^3}{\partial x^3} y(x)\right) - \left(\frac{\partial^2}{\partial x^2} y(x)\right) + x\left(\frac{\partial}{\partial x} y(x)\right) - y(x) = 0$$

Similarly in (b) the minimal order recurrence equation is

```
> fibeq := wronskeq([f(n),n!,Fib(n)],Sn,An);
```

$$fibeq := (-\mathrm{Fib}(n+1) + \mathrm{Fib}(n)\, n + \mathrm{Fib}(n))\, \mathrm{f}(n+2)$$
$$+ (\mathrm{Fib}(n+2) - 3\,\mathrm{Fib}(n)\, n - 2\,\mathrm{Fib}(n) - \mathrm{Fib}(n)\, n^2)\, \mathrm{f}(n+1)$$
$$+ (n+1)\,(\mathrm{Fib}(n+1)\, n + 2\,\mathrm{Fib}(n+1) - \mathrm{Fib}(n+2))\, \mathrm{f}(n) = 0$$

and we could plug in the explicit formula for Fibonacci numbers
  Fib(n) = (w1 ^ (n+1) - w2 ^ (n+1)) / sqrt(5)
with w1 = (1+sqrt(5))/2, w2 = -1/w1 = (1-sqrt(5))/2
to make this explicit, but that will make it uglier.
Let Sr be the algebra of shift operators with rational functions of n as coefficients.
To get a recurrence in Sr with n! and Fib(n) as solutions, we must take all the solutions of the minimal equation Fib(n) in Sr.

```
> fibeq2 := wronskeq([f(n),n!,w1^n,(-1/w1)^n],Sn,An);
```

$$fibeq2 := w1\,(-w1^2\, n - w1^2 + w1 + 3\, w1\, n + w1\, n^2 + n + 1)\, \mathrm{f}(n+3) +$$
$$(1 + n - w1 + w1^4 + w1^4\, n - 7\, w1^2 - 12\, w1^2\, n - w1^2\, n^3 - 6\, w1^2\, n^2 + w1^3)\, \mathrm{f}(n+2)$$
$$+ (6\, w1^3\, n^2 - 2 - 3\, n - 6\, w1 - 2\, w1^4 - n^2 + 6\, w1^3 - n^3\, w1 - 11\, w1\, n + 3\, w1^2$$
$$- 6\, w1\, n^2 - 3\, w1^4\, n - w1^4\, n^2 + 3\, w1^2\, n + w1^3\, n^3 + 11\, w1^3\, n + w1^2\, n^2)\, \mathrm{f}(n+1)$$
$$+ w1\,(n+1)\,(w1\, n^2 - w1^2\, n + n + 5\, w1\, n - 2\, w1^2 + 2 + 5\, w1)\, \mathrm{f}(n) = 0$$

```
> ff := ['f(n+k)'$k=0..3]:
  fibeq3 := subs(w1=(1+sqrt(5))/2,lhs(fibeq2)):
  fibeq3 := fibeq3/coeff(fibeq3,f(n+3)):
  fibeq3 := collect(fibeq3,ff,factor):
  fibeq3 := collect(numer(fibeq3),ff,factor) = 0;
```

$$fibeq3 := (n+3)\,(n+1)^2\, \mathrm{f}(n) + (5\, n + n^3 + 3 + 4\, n^2)\, \mathrm{f}(n+1)$$
$$+ (-6\, n^2 - 3 - 9\, n - n^3)\, \mathrm{f}(n+2) + n\,(n+2)\, \mathrm{f}(n+3) = 0$$

```
>
```

The new way is with LCM's. The minimal operator in the Weyl algebra annihilating sin(x) is from the first order equation
  (sin x)*y'(x) - (sin x)'*y(x) = 0, or (Dx - tan(x))*y(x)=0.
The minimal operator annihilating x is from the equation
  x*y' - x'*y = 0   or   (x Dx - 1) y = 0

```
> Isinx := Dx-tan(x); Ix := x*Dx-1; G :=
  skew_gcdex(Isinx,Ix,Dx,Ax);
```

$$Isinx := Dx - \tan(x)$$
$$Ix := x\, Dx - 1$$

```
Error, skew_gcdex expects its 1st argument, P, to be of type polynom, but received
Dx-tan(x)
```

The package won’t do this computation because it expects polynomials. If we did it by hand we would have the results given previously.
The minimal operator in the rational Weyl algebra Wr is obtained by using the minimal annihilators of the two functions in Wr. The one for sin(x) is changed to
y’’+1=0 or (Dx^2+1)y=0

```
> Isinx := Dx^2+1;
  G := skew_gcdex(Isinx,Ix,Dx,Ax);
```

$$Isinx := Dx^2 + 1$$

$$G := [x, x, -Dx, -x^2\,Dx + x, x + x\,Dx^2 - 2\,Dx]$$

```
> lc := cleanpol(skew_product(G[4],Isinx,Ax),Dx);
```

$$lc := -x^2\,Dx^3 + x\,Dx^2 - x^2\,Dx + x$$

```
> lc := cleanpol("/x,Dx);
```

$$lc := -x\,Dx^3 + Dx^2 - x\,Dx + 1$$

The minimal order equation is then

```
> applyopr(",y(x),Ax) = 0;
```

$$\mathrm{y}(x) - x\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right) + \left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) - x\left(\frac{\partial^3}{\partial x^3}\mathrm{y}(x)\right) = 0$$

For (b) we’ll have the same problem with the software for the minimal operator in the shift algebra. In the rational shift algebra, we have
(Sn-(n+1)) y(n) = 0 for y(n)=n!
(Sn^2 - Sn - 1) y(n) = 0 for y(n) = Fib(n)

```
> I1 := Sn-(n+1): I2 := Sn^2-Sn-1: G :=
  skew_gcdex(I1,I2,Sn,An);
```

$$G := [2\,n + n^2, -1 - Sn - n, 1, -4\,n - n^2 - 3 - 3\,Sn\,n - 3\,Sn - n^2\,Sn + 2\,Sn^2\,n + n^2\,Sn^2, 3 + n^3 + 7\,n + 5\,n^2 - 2\,Sn\,n - n^2\,Sn]$$

The minimal operator in Sr annihilating n! and Fib(n) is

```
> lc := cleanpol(skew_product(G[4],I1,An),Sn);
```

$$lc := (n+2)\,Sn^3\,n + (-n^3 - 6\,n^2 - 9\,n - 3)\,Sn^2 + (n^3 + 4\,n^2 + 5\,n + 3)\,Sn + (n+3)\,(n+1)^2$$

and in function notation this is

```
> applyopr(lc,a(n),An) = 0;
```

$$(n+3)\,(n+1)^2\,\mathrm{a}(n) + (n^3 + 4\,n^2 + 5\,n + 3)\,\mathrm{a}(n+1) + (-n^3 - 6\,n^2 - 9\,n - 3)\,\mathrm{a}(n+2) + n\,(n+2)\,\mathrm{a}(n+3) = 0$$

```
>
> fn := 5*n^3 + 4*n^2;
```

$$fn := 5\,n^3 + 4\,n^2$$

```
> expand(applyopr(Sn-1,fn,An));
```

$$15\,n^2 + 23\,n + 9$$

```
> expand(applyopr(Sn-1,",An));
```

$$38 + 30\,n$$

```
> expand(applyopr(Sn-1,",An));
```

$$30$$

```
> FF := 9/2*ff(n,2) + 38/6*ff(n,3) + 30/24*ff(n,4);
```

$$FF := \frac{9}{2}\,n\,(n-1) + \frac{19}{3}\,n\,(n-1)\,(n-2) + \frac{5}{4}\,n\,(n-1)\,(n-2)\,(n-3)$$

```
> expand(FF);
```

$$-\frac{3}{4}\,n^2 + \frac{2}{3}\,n - \frac{7}{6}\,n^3 + \frac{5}{4}\,n^4$$

```
> expand(subs(n=m+1,"));
```

$$\frac{13}{4}\,m^2 + \frac{2}{3}\,m + \frac{23}{6}\,m^3 + \frac{5}{4}\,m^4$$

```
>
```